
1. Overview

This document lists all changes needed to correct problems in the TTCN implementation of test case 7.1.3.13a which is part of the LTE/SAE test suite in the ‘iwd-TTCN3-B2013-03_D15wk10’ ATS delivery.

The test case can be demonstrated to run with one LTE UE (see section 5). Execution log is provided as evidence.

3. Verification Test Summary

Test Case:
TC_7_1_3_13a
Test Group:
LTE_A\7_1\MAC_SSC.ttcn
ATS Version:
iwd-TTCN3-B2013-03_D15wk10
System Simulator used:
R&S CMW500 
UE used:
Qualcomm MDM 9x35
Verification Status:
PASS


4. Corrections required for TC 7.1.3.13a
4.1  Corrections to function ‘f_TC_7_1_3_13a_EUTRA’
	Function name
	f_TC_7_1_3_13a_EUTRA ()

	Reason for change
	1.   For Transmission Mode 7 with DCI 1,antenna port 5 has to be used as per TS 36.213 Table 7.1-5.  
2.   DCI 1 is to be used for sending PDSCH data (see TC requirements).

	Summary of change
	1.   Corrected the antenna port number to 5.
2.   Configured and passed the correct DCI format to be used.

	TTCN module
	LTE_A\7_1\MAC_SSC.ttcn

	MCC160 Comment
	


Before change

...

function f_TC_7_1_3_13a_EUTRA() runs on EUTRA_PTC

  { /* L2/MAC

     * 7.1.3.13a : TDD additional special subframe configuration / Special subframe pattern 7 with Extended Cyclic Prefix / UE-specific reference signals based transmission scheme

     */

    f_TC_7_1_3_13_Common(ssp4, ssp7, len2, tm7, 7)

  };
... 
After change

... 

  function f_TC_7_1_3_13a_EUTRA() runs on EUTRA_PTC

  { /* L2/MAC

     * 7.1.3.13a : TDD additional special subframe configuration / Special subframe pattern 7 with Extended Cyclic Prefix / UE-specific reference signals based transmission scheme

     */

    f_TC_7_1_3_13_Common(ssp4, ssp7, len2, tm7, 5, dci_1)  
  };
... 

4.2  Corrections to function ‘f_TC_7_1_3_13_Common’
	Function name
	f_TC_7_1_3_13_Common ()

	Reason for change
	1.   Cylic Prefix length as required by the TC is not configured correctly in SS.
2.   Use of DCI 2B (--> TC 7.1.3.13) is hardcoded in the Common TC function for sending PDSCH data.

3.   When using DCI 1 and TM7 it is a single antenna port scheme as per TS 36.213 Table 7.1-5.

	Summary of change
	1.   Corrected the implementation to configure the CP length in the SS.

2.   Configured the DCI format as per TC requirements.

3.   Corrected the transmission scheme as per TC requirements.

	TTCN module
	LTE_A\7_1\MAC_SSC.ttcn

	MCC160 Comment
	


Before change

...

function f_TC_7_1_3_13_Common(TDD_Config.specialSubframePatterns p_SpecialSubframePatterns,

                                TDD_Config_v1130.specialSubframePatterns_v1130 p_SpecialSubframePatterns_v1130,

                                UL_CyclicPrefixLength p_UL_CyclicPrefixLength,

                                AntennaInfoDedicated.transmissionMode p_TransmissionMode,

                                AntennaPortId_Type p_AntennaPortId_Type) runs on EUTRA_PTC

  { /* L2/MAC

     * TDD additional special subframe configuration / Special subframe pattern 9/7 / UE-specific reference signals based transmission scheme

     */

    var Dl_Bandwidth_Type v_Dl_Bandwidth_Cell1;

    var SubFrameTiming_Type v_Timing;

    var octetstring v_EncodedPdcpPdu;

    var octetstring v_EncodedRlcPdu;

    var EUTRA_CellInfo_Type v_CellInfo_Cell1;

    var EUTRA_FDD_TDD_Mode_Type v_EUTRA_FDD_TDD_Mode; //@sic R5s141331 sic@

    timer t_Watchdog := 5.0;

    f_EUTRA_Init (c1);

    //Test applicable only for TDD

    v_EUTRA_FDD_TDD_Mode := f_EUTRA_CellInfo_GetFDD_TDD(eutra_Cell1); //@sic R5s141331 sic@

    if (v_EUTRA_FDD_TDD_Mode == FDD) {

        FatalError(__FILE__, __LINE__, "Invalid operation mode");

    }

    f_EUTRA_CellInfo_SetMIMO_2TX(eutra_Cell1);//@sic R5s141239 sic@
    v_Dl_Bandwidth_Cell1 := f_EUTRA_CellInfo_GetDL_ChBandwidth(eutra_Cell1);

    //set specialSubframePatterns to ssp9 for normal cyclic prefix for cell 1

    f_EUTRA_CellInfo_SetSysInfo_TDD_Config_v1130(eutra_Cell1,p_SpecialSubframePatterns,p_SpecialSubframePatterns_v1130);

    f_EUTRA_CellInfo_SetSysInfo_UL_CyclicPrefixLength(eutra_Cell1,p_UL_CyclicPrefixLength);

    f_EUTRA_CellConfig_Def (eutra_Cell1);

    // configure SS for TDD candidates selection in special sub frames

    f_EUTRA_SS_CommonCellConfig(eutra_Cell1, cas_PhysicalLayerConfigDL_CcchDcchDtchDL_Config_REQ(eutra_Cell1, cs_TimingInfo_Now, cs_PhysicalLayerConfigDL_CCESpecialSubframe_TDD(v_Dl_Bandwidth_Cell1)));

    v_CellInfo_Cell1 := f_EUTRA_CellInfo_Get(eutra_Cell1);

    f_EUTRA_Preamble (eutra_Cell1, STATE2A_TESTLOOP_ModeA);

    f_EUTRA_AS_CipheringAlgorithm_Set(eea0); // set null AS ciphering algorithm

    f_EUTRA_CellInfo_SetAntennaInfo(eutra_Cell1, p_TransmissionMode);

    //f_EUTRA_SS_CommonCellConfig(eutra_Cell1, cas_PhysicalLayerConfigDL_CcchDcchDtchDL_Config_REQ(eutra_Cell1, cs_TimingInfo_Now, cs_PhysicalLayerConfigDL_AntennaInfo(cs_DownlinkAntennaGroupConfig(v_CellInfo_Cell1.AntennaInfo.SS_AntennaInfoCommon, 5))));

    f_EUTRA_LoopBackActivation_State4_TA_Infinity_LB_NoScaling_AntennaInfo_ExplicitValue(-, -, v_CellInfo_Cell1.AntennaInfo.SS_AntennaInfoCommon, p_AntennaPortId_Type);

    f_Delay(0.660); // 600 ms previous time alignment timer

    ///////////////////////////////////////////////////////////////////////////////////

    //      Test Body

    //////////////////////////////////////////////////////////////////////////////////

    f_EUTRA_TestBody_Set(true, layer2);

    f_SS_Config1TM_DRB_NoMACHeader(eutra_Cell1, tsc_RbId_DRB1, cs_MAC_TestMode_NoHeaderManipulationDLOnly);

    // configure SS for default UL grant allocation at SR reception but no automatic PUCCH synch as time alignment timer is infinit;

    f_EUTRA_StartULGrantTransmissionOnSR_Reception_No_PUCCH_Sync(eutra_Cell1, cs_TimingInfo_Now, cs_DciInfo_CcchDcchDtchUL_Def);

    // to configure SS to report reception of HARQ ACK or NACK

    f_SS_ConfigUL_HARQ_IndMode (eutra_Cell1, tsc_L1Mac_IndicationMode_Enable);

    // to configure SS to report reception of SR

    f_SS_ConfigSchedulingRequestIndMode(eutra_Cell1, tsc_L1Mac_IndicationMode_Enable);

    //@siclog "Steps 1" siclog@

    // The SS indicates a new transmission on PDCCH and transmits a MAC PDU in SF-Num 1 or 6 where the DwPTS belongs to.

    // 13B PDCP SDU + 2B PDCP H + 2B RLC Header +1 B MAC SH  = 18 B= 144 bits

    v_EncodedPdcpPdu  := f_PDCP_PDU_Encvalue(cs_PDCP_PDU_LongSN(cs_PDCP_DataPdu_LongSN(tsc_B12_SQN0, crs_PDCP_SDU_13B)));

    v_EncodedRlcPdu   := f_RLC_PDU_Encvalue(cs_RLC_AMD_1PDU(tsc_NoPoll, int2bit(0, 10), v_EncodedPdcpPdu));

    f_EUTRA_SS_CommonCellConfig(eutra_Cell1, cas_CcchDcchDtchDL_Config_REQ(eutra_Cell1,

                                                                           cs_TimingInfo_Now,

                                                                           cs_DciDLInfo_Explicit(cs_DciDlInfoExplicit(2,//Imcs=2,Nprb=3

                                                                                                                      dci_2B,

                                                                                                                      ra_0,

                                                                                                                      8))));  //N'prb, Nprb=max{floor(N'prb*0.375),1}

    v_Timing := f_EUTRA_GetNextSendOccasion(eutra_Cell1);

    DRB.send (cas_DRB_COMMOM_REQ_MAC_PDU(eutra_Cell1,

                                         tsc_RbId_DRB1,

                                         cs_TimingInfo(v_Timing.SFN.Number, 1),

                                         cs_MAC_PDUList_1 (cs_MAC_PDU_NoPadding_RLCPDU_LCID(tsc_LchId_DRB1,

                                                                                            v_EncodedRlcPdu))));

    // prepare for sending status PDU to avoid any retransmissions

    // MAC SH 2B(RLCSTATUS) + SH 1B(PADING) + RLC STATUS 2B + PADING 6B = 11B = 88bit

    f_EUTRA_SS_CommonCellConfig(eutra_Cell1, cas_CcchDcchDtchDL_Config_REQ(eutra_Cell1,

                                                                           cs_TimingInfo(v_Timing.SFN.Number, 4),

                                                                           cs_DciDLInfo_Explicit(cs_DciDlInfoExplicit(1,//Imcs=1,Nprb=3

                                                                                                                      dci_2B,

                                                                                                                      ra_0,

                                                                                                                      8)))); //N'prb, Nprb=max{floor(N'prb*0.375),1}
... 
After change

... 

  function f_TC_7_1_3_13_Common(TDD_Config.specialSubframePatterns p_SpecialSubframePatterns,

                                TDD_Config_v1130.specialSubframePatterns_v1130 p_SpecialSubframePatterns_v1130,

                                UL_CyclicPrefixLength p_UL_CyclicPrefixLength,

                                AntennaInfoDedicated.transmissionMode p_TransmissionMode,

                                AntennaPortId_Type p_AntennaPortId_Type,

                                PdcchDciFormat_Type p_dciFomat) runs on EUTRA_PTC  

  { /* L2/MAC

     * TDD additional special subframe configuration / Special subframe pattern 9/7 / UE-specific reference signals based transmission scheme

     */

    var Dl_Bandwidth_Type v_Dl_Bandwidth_Cell1;

    var SubFrameTiming_Type v_Timing;

    var octetstring v_EncodedPdcpPdu;

    var octetstring v_EncodedRlcPdu;

    var EUTRA_CellInfo_Type v_CellInfo_Cell1;

    var EUTRA_FDD_TDD_Mode_Type v_EUTRA_FDD_TDD_Mode; //@sic R5s141331 sic@

    var EUTRA_CyclicPrefix_Type v_cyclicPrefixLength := normal;  
    if(p_UL_CyclicPrefixLength == len2) {

      v_cyclicPrefixLength := extended ; 

    }
    timer t_Watchdog := 5.0;

    f_EUTRA_Init (c1);

    //Test applicable only for TDD

    v_EUTRA_FDD_TDD_Mode := f_EUTRA_CellInfo_GetFDD_TDD(eutra_Cell1); //@sic R5s141331 sic@

    if (v_EUTRA_FDD_TDD_Mode == FDD) {

        FatalError(__FILE__, __LINE__, "Invalid operation mode");

    }

    if (p_TransmissionMode == tm8_v920) {

        f_EUTRA_CellInfo_SetMIMO_2TX(eutra_Cell1);//@sic R5s141239 sic@

    }
    v_Dl_Bandwidth_Cell1 := f_EUTRA_CellInfo_GetDL_ChBandwidth(eutra_Cell1);

    //set specialSubframePatterns to ssp9 for normal cyclic prefix for cell 1

    f_EUTRA_CellInfo_SetSysInfo_TDD_Config_v1130(eutra_Cell1,p_SpecialSubframePatterns,p_SpecialSubframePatterns_v1130);

    f_EUTRA_CellInfo_SetSysInfo_UL_CyclicPrefixLength(eutra_Cell1,p_UL_CyclicPrefixLength);

    f_EUTRA_CellConfig_Def (eutra_Cell1,-,v_cyclicPrefixLength);  
    // configure SS for TDD candidates selection in special sub frames

    f_EUTRA_SS_CommonCellConfig(eutra_Cell1, cas_PhysicalLayerConfigDL_CcchDcchDtchDL_Config_REQ(eutra_Cell1, cs_TimingInfo_Now, cs_PhysicalLayerConfigDL_CCESpecialSubframe_TDD(v_Dl_Bandwidth_Cell1)));

    v_CellInfo_Cell1 := f_EUTRA_CellInfo_Get(eutra_Cell1);

    f_EUTRA_Preamble (eutra_Cell1, STATE2A_TESTLOOP_ModeA);

    f_EUTRA_AS_CipheringAlgorithm_Set(eea0); // set null AS ciphering algorithm

    f_EUTRA_CellInfo_SetAntennaInfo(eutra_Cell1, p_TransmissionMode);

    //f_EUTRA_SS_CommonCellConfig(eutra_Cell1, cas_PhysicalLayerConfigDL_CcchDcchDtchDL_Config_REQ(eutra_Cell1, cs_TimingInfo_Now, cs_PhysicalLayerConfigDL_AntennaInfo(cs_DownlinkAntennaGroupConfig(v_CellInfo_Cell1.AntennaInfo.SS_AntennaInfoCommon, 5))));

    f_EUTRA_LoopBackActivation_State4_TA_Infinity_LB_NoScaling_AntennaInfo_ExplicitValue(-, -, v_CellInfo_Cell1.AntennaInfo.SS_AntennaInfoCommon, p_AntennaPortId_Type);

    f_Delay(0.660); // 600 ms previous time alignment timer

    ///////////////////////////////////////////////////////////////////////////////////

    //      Test Body

    //////////////////////////////////////////////////////////////////////////////////

    f_EUTRA_TestBody_Set(true, layer2);

    f_SS_Config1TM_DRB_NoMACHeader(eutra_Cell1, tsc_RbId_DRB1, cs_MAC_TestMode_NoHeaderManipulationDLOnly);

    // configure SS for default UL grant allocation at SR reception but no automatic PUCCH synch as time alignment timer is infinit;

    f_EUTRA_StartULGrantTransmissionOnSR_Reception_No_PUCCH_Sync(eutra_Cell1, cs_TimingInfo_Now, cs_DciInfo_CcchDcchDtchUL_Def);

    // to configure SS to report reception of HARQ ACK or NACK

    f_SS_ConfigUL_HARQ_IndMode (eutra_Cell1, tsc_L1Mac_IndicationMode_Enable);

    // to configure SS to report reception of SR

    f_SS_ConfigSchedulingRequestIndMode(eutra_Cell1, tsc_L1Mac_IndicationMode_Enable);

    //@siclog "Steps 1" siclog@

    // The SS indicates a new transmission on PDCCH and transmits a MAC PDU in SF-Num 1 or 6 where the DwPTS belongs to.

    // 13B PDCP SDU + 2B PDCP H + 2B RLC Header +1 B MAC SH  = 18 B= 144 bits

    v_EncodedPdcpPdu  := f_PDCP_PDU_Encvalue(cs_PDCP_PDU_LongSN(cs_PDCP_DataPdu_LongSN(tsc_B12_SQN0, crs_PDCP_SDU_13B)));

    v_EncodedRlcPdu   := f_RLC_PDU_Encvalue(cs_RLC_AMD_1PDU(tsc_NoPoll, int2bit(0, 10), v_EncodedPdcpPdu));

    f_EUTRA_SS_CommonCellConfig(eutra_Cell1, cas_CcchDcchDtchDL_Config_REQ(eutra_Cell1,

                                                                           cs_TimingInfo_Now,

                                                                           cs_DciDLInfo_Explicit(cs_DciDlInfoExplicit(2,//Imcs=2,Nprb=3

                                                                                                                      p_dciFomat, 

                                                                                                                      ra_0,

                                                                                                                      8))));  //N'prb, Nprb=max{floor(N'prb*0.375),1}

    v_Timing := f_EUTRA_GetNextSendOccasion(eutra_Cell1);

    DRB.send (cas_DRB_COMMOM_REQ_MAC_PDU(eutra_Cell1,

                                         tsc_RbId_DRB1,

                                         cs_TimingInfo(v_Timing.SFN.Number, 1),

                                         cs_MAC_PDUList_1 (cs_MAC_PDU_NoPadding_RLCPDU_LCID(tsc_LchId_DRB1,

                                                                                            v_EncodedRlcPdu))));

    // prepare for sending status PDU to avoid any retransmissions

    // MAC SH 2B(RLCSTATUS) + SH 1B(PADING) + RLC STATUS 2B + PADING 6B = 11B = 88bit

    f_EUTRA_SS_CommonCellConfig(eutra_Cell1, cas_CcchDcchDtchDL_Config_REQ(eutra_Cell1,

                                                                           cs_TimingInfo(v_Timing.SFN.Number, 4),

                                                                           cs_DciDLInfo_Explicit(cs_DciDlInfoExplicit(1,//Imcs=1,Nprb=3

                                                                                                                      p_dciFomat, 

                                                                                                                      ra_0,

                                                                                                                      8)))); //N'prb, Nprb=max{floor(N'prb*0.375),1}

... 

4.3  Corrections to function ‘f_EUTRA_CellConfig_Def’ 
	Function name
	f_EUTRA_CellConfig_Def ()

	Reason for change
	See 4.2 change 1 above.

	Summary of change
	See implementation below.

	TTCN module
	Common\EUTRA\EUTRA_ConfigurationSteps.ttcn

	MCC160 Comment
	


Before change

...

function f_EUTRA_CellConfig_Def(EUTRA_CellId_Type p_CellId,

                                  template (omit) CA_CellInitialConfig_Type p_CA_CellInitialConfig := omit) runs on EUTRA_PTC

  {

    f_EUTRA_SS_ConfigureActiveCell(p_CellId, p_CA_CellInitialConfig);

    f_EUTRA_SS_ConfigureSRBs(p_CellId);

    f_EUTRA_SS_DRB1_DefConfig(p_CellId);
  }
... 
After change

...
function f_EUTRA_CellConfig_Def(EUTRA_CellId_Type p_CellId,

                                  template (omit) CA_CellInitialConfig_Type p_CA_CellInitialConfig := omit,

                                  EUTRA_CyclicPrefix_Type p_cyclicPrefixLength := normal)  runs on EUTRA_PTC     
  {

    f_EUTRA_SS_ConfigureActiveCell(p_CellId, p_CA_CellInitialConfig, p_cyclicPrefixLength);      f_EUTRA_SS_ConfigureSRBs(p_CellId);

    f_EUTRA_SS_DRB1_DefConfig(p_CellId);

  }
... 

4.4 Corrections to function ‘f_EUTRA_SS_ConfigureActiveCell’ 
	Function name
	f_EUTRA_SS_ConfigureActiveCell ()

	Reason for change
	See 4.2 change 1 above.

	Summary of change
	See implementation below.

	TTCN module
	Common\EUTRA\EUTRA_ConfigurationSteps.ttcn

	MCC160 Comment
	


Before change

...

function f_EUTRA_SS_ConfigureActiveCell(EUTRA_CellId_Type p_CellId,

                                          template (omit) CA_CellInitialConfig_Type p_CA_CellInitialConfig := omit) runs on EUTRA_PTC

  {

    var EUTRA_CellInfo_Type v_CellInfo := f_EUTRA_CellInfo_Get (p_CellId);

    var template (value) AllSiSchedul_Type    v_SiSchedul;

    var template (value) ContentionResolution_ContainedRlcPdu_Type v_ContentionResolution_ContainedRlcPdu;

    var template (value) DL_CCCH_Message v_RrcConnSetup;

    var octetstring v_EncodedRrcConnSetup;

    //Get FDD or TDD mode from Cell-configuration

    var EUTRA_FDD_TDD_Mode_Type  v_EUTRA_FDD_TDD_Mode := f_EUTRA_CellInfo_GetFDD_TDD (p_CellId);

    var template (value) StaticCellInfo_Type v_StaticCellInfo;

    var template (omit) TDD_Config_v1130 v_TDD_Config_v1130 := omit;

    v_SiSchedul := f_EUTRA_SS_SetupSchedulingInfo(v_CellInfo.Sysinfo.BCCH_Info.SIB1.message_.c1.systemInformationBlockType1, v_EUTRA_FDD_TDD_Mode);      /* derive SS's scheduling information from SIB1 */

    select (v_EUTRA_FDD_TDD_Mode) {

      case (FDD) {

        if (f_EUTRA_CellInfo_GetNoUL_Frequency(p_CellId)) {

          // UL frequency NOT defined in the band @sic R5-134786 sic@

          v_StaticCellInfo := cs_StaticCellInfo_FDD(v_CellInfo.CellIds.PhysicalCellId,

                                                    v_CellInfo.Frequency.FrequencyBandIndicator,

                                                    v_CellInfo.Frequency.UL_DL_Earfcn.dl_CarrierFreq,

                                                    v_CellInfo.Frequency.DL_ChBandwidth,

                                                    v_CellInfo.CellTimingInfo.Tcell,       /* @sic R5s100029 change 3 sic@ */

                                                    v_CellInfo.CellTimingInfo.SfnOffset,

                                                    omit);

        } else {

            // UL frequency defined in the band

          v_StaticCellInfo := cs_StaticCellInfo_FDD(v_CellInfo.CellIds.PhysicalCellId,

                                                    v_CellInfo.Frequency.FrequencyBandIndicator,

                                                    v_CellInfo.Frequency.UL_DL_Earfcn.dl_CarrierFreq,

                                                    v_CellInfo.Frequency.DL_ChBandwidth,

                                                    v_CellInfo.CellTimingInfo.Tcell,       /* @sic R5s100029 change 3 sic@ */

                                                    v_CellInfo.CellTimingInfo.SfnOffset,

                                                    cs_UplinkStaticCellInfo(     // @sic R5134786 sic@

                                                                            v_CellInfo.Frequency.UL_DL_Earfcn.ul_CarrierFreq,

                                                                            v_CellInfo.Frequency.UL_ChBandwidth));  /* @sic R5s100029 change 3 sic@ */

        }

      }

      case (TDD) {

        if(ispresent(v_CellInfo.Sysinfo.BCCH_Info.SIB1.message_.c1.systemInformationBlockType1.nonCriticalExtension.nonCriticalExtension.nonCriticalExtension.tdd_Config_v1130)){

          v_TDD_Config_v1130 := v_CellInfo.Sysinfo.BCCH_Info.SIB1.message_.c1.systemInformationBlockType1.nonCriticalExtension.nonCriticalExtension.nonCriticalExtension.tdd_Config_v1130;

        }

        v_StaticCellInfo := cs_StaticCellInfo_TDD (v_CellInfo.CellIds.PhysicalCellId,

                                                   v_CellInfo.Frequency.FrequencyBandIndicator,

                                                   v_CellInfo.Frequency.UL_DL_Earfcn.dl_CarrierFreq,

                                                   v_CellInfo.Frequency.DL_ChBandwidth, // @sic R5s100190 sic@

                                                   v_CellInfo.CellTimingInfo.Tcell,   // @sic R5s100513 sic@

                                                   v_CellInfo.CellTimingInfo.SfnOffset,   // @sic R5s100513 sic@

                                                   v_CellInfo.Sysinfo.BCCH_Info.SIB1.message_.c1.systemInformationBlockType1.tdd_Config,

                                                   v_TDD_Config_v1130);

      }

    }

    if (f_EUTRA_DlCcchMsgInSeparateMacPdu()) {     /* @sic R5-120721 change 3 sic@ */

      v_ContentionResolution_ContainedRlcPdu := cs_ContentionResolution_RlcPdu_None;

    } else {

      v_RrcConnSetup := f_EUTRA_508_RRCConnectionSetup(p_CellId);

      v_EncodedRrcConnSetup := f_RRC_DL_CCCH_Encvalue(v_RrcConnSetup);

      v_ContentionResolution_ContainedRlcPdu := cs_ContentionResolution_RlcPdu(v_EncodedRrcConnSetup);

    }

    if (f_EUTRA_CellInfo_GetNoUL_Frequency(p_CellId)) {

        // UL frequency NOT defined in the band @sic R5-134786 sic@

        f_EUTRA_SS_CommonCellConfig(p_CellId, cas_CellConfig_NoULFreq_REQ(p_CellId,

                                                                          cs_TimingInfo_Now,

                                                                          v_StaticCellInfo,

                                                                          v_SiSchedul,

                                                                          v_CellInfo,

                                                                          p_CA_CellInitialConfig));

    } else {

        // UL frequency defined in the band @sic R5-134786 sic@

        f_EUTRA_SS_CommonCellConfig(p_CellId, cas_CellConfig_Def_REQ(p_CellId,

                                                                 cs_TimingInfo_Now,

                                                                 v_StaticCellInfo,

                                                                 v_SiSchedul,

                                                                 v_CellInfo,

                                                                 v_ContentionResolution_ContainedRlcPdu,

                                                                 p_CA_CellInitialConfig));

    }

    if (f_TestcaseIsL2Testcase() or f_TestcaseIsDrbTestcase()) {     /* @sic R5s100135, R5-106675 change 1 sic@ */

      f_EUTRA_SS_CommonL1MacIndCtrlConfig(p_CellId, cas_HarqErrorIndication_REQ(p_CellId));

    }

  }
... 
After change

... 

function f_EUTRA_SS_ConfigureActiveCell(EUTRA_CellId_Type p_CellId,

                                          template (omit) CA_CellInitialConfig_Type p_CA_CellInitialConfig := omit,

                                          EUTRA_CyclicPrefix_Type p_cyclicPrefixLength := normal) runs on EUTRA_PTC 
{

    var EUTRA_CellInfo_Type v_CellInfo := f_EUTRA_CellInfo_Get (p_CellId);

    var template (value) AllSiSchedul_Type    v_SiSchedul;

    var template (value) ContentionResolution_ContainedRlcPdu_Type v_ContentionResolution_ContainedRlcPdu;

    var template (value) DL_CCCH_Message v_RrcConnSetup;

    var octetstring v_EncodedRrcConnSetup;

    //Get FDD or TDD mode from Cell-configuration

    var EUTRA_FDD_TDD_Mode_Type  v_EUTRA_FDD_TDD_Mode := f_EUTRA_CellInfo_GetFDD_TDD (p_CellId);

    var template (value) StaticCellInfo_Type v_StaticCellInfo;

    var template (omit) TDD_Config_v1130 v_TDD_Config_v1130 := omit;

    v_SiSchedul := f_EUTRA_SS_SetupSchedulingInfo(v_CellInfo.Sysinfo.BCCH_Info.SIB1.message_.c1.systemInformationBlockType1, v_EUTRA_FDD_TDD_Mode);      /* derive SS's scheduling information from SIB1 */

    select (v_EUTRA_FDD_TDD_Mode) {

      case (FDD) {

        if (f_EUTRA_CellInfo_GetNoUL_Frequency(p_CellId)) {

          // UL frequency NOT defined in the band @sic R5-134786 sic@

          v_StaticCellInfo := cs_StaticCellInfo_FDD(v_CellInfo.CellIds.PhysicalCellId,

                                                    v_CellInfo.Frequency.FrequencyBandIndicator,

                                                    v_CellInfo.Frequency.UL_DL_Earfcn.dl_CarrierFreq,

                                                    v_CellInfo.Frequency.DL_ChBandwidth,

                                                    v_CellInfo.CellTimingInfo.Tcell,       /* @sic R5s100029 change 3 sic@ */

                                                    v_CellInfo.CellTimingInfo.SfnOffset,

                                                    omit);

        } else {

            // UL frequency defined in the band

          v_StaticCellInfo := cs_StaticCellInfo_FDD(v_CellInfo.CellIds.PhysicalCellId,

                                                    v_CellInfo.Frequency.FrequencyBandIndicator,

                                                    v_CellInfo.Frequency.UL_DL_Earfcn.dl_CarrierFreq,

                                                    v_CellInfo.Frequency.DL_ChBandwidth,

                                                    v_CellInfo.CellTimingInfo.Tcell,       /* @sic R5s100029 change 3 sic@ */

                                                    v_CellInfo.CellTimingInfo.SfnOffset,

                                                    cs_UplinkStaticCellInfo(     // @sic R5134786 sic@

                                                                            v_CellInfo.Frequency.UL_DL_Earfcn.ul_CarrierFreq,

                                                                            v_CellInfo.Frequency.UL_ChBandwidth));  /* @sic R5s100029 change 3 sic@ */

        }

      }

      case (TDD) {

        if(ispresent(v_CellInfo.Sysinfo.BCCH_Info.SIB1.message_.c1.systemInformationBlockType1.nonCriticalExtension.nonCriticalExtension.nonCriticalExtension.tdd_Config_v1130)){

          v_TDD_Config_v1130 := v_CellInfo.Sysinfo.BCCH_Info.SIB1.message_.c1.systemInformationBlockType1.nonCriticalExtension.nonCriticalExtension.nonCriticalExtension.tdd_Config_v1130;

        }

        v_StaticCellInfo := cs_StaticCellInfo_TDD (v_CellInfo.CellIds.PhysicalCellId,

                                                   v_CellInfo.Frequency.FrequencyBandIndicator,

                                                   v_CellInfo.Frequency.UL_DL_Earfcn.dl_CarrierFreq,

                                                   v_CellInfo.Frequency.DL_ChBandwidth, // @sic R5s100190 sic@

                                                   v_CellInfo.CellTimingInfo.Tcell,   // @sic R5s100513 sic@

                                                   v_CellInfo.CellTimingInfo.SfnOffset,   // @sic R5s100513 sic@

                                                   v_CellInfo.Sysinfo.BCCH_Info.SIB1.message_.c1.systemInformationBlockType1.tdd_Config,

                                                   v_TDD_Config_v1130,

                                                   p_cyclicPrefixLength);
      }

    }

    if (f_EUTRA_DlCcchMsgInSeparateMacPdu()) {     /* @sic R5-120721 change 3 sic@ */

      v_ContentionResolution_ContainedRlcPdu := cs_ContentionResolution_RlcPdu_None;

    } else {

      v_RrcConnSetup := f_EUTRA_508_RRCConnectionSetup(p_CellId);

      v_EncodedRrcConnSetup := f_RRC_DL_CCCH_Encvalue(v_RrcConnSetup);

      v_ContentionResolution_ContainedRlcPdu := cs_ContentionResolution_RlcPdu(v_EncodedRrcConnSetup);

    }

    if (f_EUTRA_CellInfo_GetNoUL_Frequency(p_CellId)) {

        // UL frequency NOT defined in the band @sic R5-134786 sic@

        f_EUTRA_SS_CommonCellConfig(p_CellId, cas_CellConfig_NoULFreq_REQ(p_CellId,

                                                                          cs_TimingInfo_Now,

                                                                          v_StaticCellInfo,

                                                                          v_SiSchedul,

                                                                          v_CellInfo,

                                                                          p_CA_CellInitialConfig));

    } else {

        // UL frequency defined in the band @sic R5-134786 sic@

        f_EUTRA_SS_CommonCellConfig(p_CellId, cas_CellConfig_Def_REQ(p_CellId,

                                                                 cs_TimingInfo_Now,

                                                                 v_StaticCellInfo,

                                                                 v_SiSchedul,

                                                                 v_CellInfo,

                                                                 v_ContentionResolution_ContainedRlcPdu,

                                                                 p_CA_CellInitialConfig));

    }

    if (f_TestcaseIsL2Testcase() or f_TestcaseIsDrbTestcase()) {     /* @sic R5s100135, R5-106675 change 1 sic@ */

      f_EUTRA_SS_CommonL1MacIndCtrlConfig(p_CellId, cas_HarqErrorIndication_REQ(p_CellId));

    }

  }
... 

4.5 Corrections to template ‘cs_StaticCellInfo_TDD’ 
	Template name
	cs_StaticCellInfo_TDD ()

	Reason for change
	See 4.2 change 1 above.

	Summary of change
	See implementation below.

	TTCN module
	Common\EUTRA_Templates\EUTRA_CellCfg_Templates.ttcn

	MCC160 Comment
	


Before change

...

template (value) StaticCellInfo_Type cs_StaticCellInfo_TDD(PhysCellId    p_PhysicalCellIdentity,

                                                             FreqBandIndicator p_EutraBand,

                                                             ARFCN_ValueEUTRA  p_EarfcnDL,

                                                             Dl_Bandwidth_Type p_Dl_Bandwidth, // @sic R5s100190 sic@

                                                             integer           p_Tcell, // @sic R5s100513 sic@

                                                             integer           p_SfnOffset, // @sic R5s100513 sic@

                                                             TDD_Config        p_TDD_Config,

                                                             template (omit) TDD_Config_v1130 p_TDD_Config_v1130 := omit) :=

  { /* @status    APPROVED (IMS, IMS_IRAT, LTE, LTE_A, LTE_IRAT, POS) */

    Common := {

      RAT := {

        TDD := {

          Configuration := cs_TDD_Config_r8(p_TDD_Config,p_TDD_Config_v1130)

        }

      },

      PhysicalCellId := p_PhysicalCellIdentity,

      eNB_CellId     := omit,

      EutraBand      := p_EutraBand,

      CellTimingInfo := {

        Tcell := p_Tcell, // @sic R5s100513 sic@

        SfnOffset := p_SfnOffset // @sic R5s100513 sic@

      }

    },

    Downlink := {

      Earfcn       := p_EarfcnDL,

      Bandwidth    := p_Dl_Bandwidth,             // @sic R5s100190 sic@

      RBSize       := n_RB_SC_12,                 /* normal sub-frequency spacing */

      CyclicPrefix := normal
    },

    Uplink := omit

  };
... 
After change

... 

template (value) StaticCellInfo_Type cs_StaticCellInfo_TDD(PhysCellId    p_PhysicalCellIdentity,

                                                             FreqBandIndicator p_EutraBand,

                                                             ARFCN_ValueEUTRA  p_EarfcnDL,

                                                             Dl_Bandwidth_Type p_Dl_Bandwidth, // @sic R5s100190 sic@

                                                             integer           p_Tcell, // @sic R5s100513 sic@

                                                             integer           p_SfnOffset, // @sic R5s100513 sic@

                                                             TDD_Config        p_TDD_Config,

                                                             template (omit) TDD_Config_v1130 p_TDD_Config_v1130 := omit,

                                                             EUTRA_CyclicPrefix_Type p_cyclicPrefixLength := normal) :=  

  { /* @status    APPROVED (IMS, IMS_IRAT, LTE, LTE_A, LTE_IRAT, POS) */

    Common := {

      RAT := {

        TDD := {

          Configuration := cs_TDD_Config_r8(p_TDD_Config,p_TDD_Config_v1130)

        }

      },

      PhysicalCellId := p_PhysicalCellIdentity,

      eNB_CellId     := omit,

      EutraBand      := p_EutraBand,

      CellTimingInfo := {

        Tcell := p_Tcell, // @sic R5s100513 sic@

        SfnOffset := p_SfnOffset // @sic R5s100513 sic@

      }

    },

    Downlink := {

      Earfcn       := p_EarfcnDL,

      Bandwidth    := p_Dl_Bandwidth,             // @sic R5s100190 sic@

      RBSize       := n_RB_SC_12,                 /* normal sub-frequency spacing */

      CyclicPrefix := p_cyclicPrefixLength
    },

    Uplink := omit

  };
... 

5. Execution Log Files

5.1 Qualcomm MDM 9x35
The Qualcomm MDM 9x35 UE passed this LTE-TDD MAC test case on R&S CMW500 in eTDD band 38. The documentation below is enclosed as evidence of the successful test case run [1]:

· Test Case Execution log file: 
Rohde-Schwarz\tc_7_1_3_13a_LTE_TDD_Qualcomm.log   
(Note: PICS/PIXIT settings are captured at the beginning of the TLI log)

In the log file (in .txt / -html format) the complete test case execution can be seen. All TLI events are presented and message contents are fully decoded and can be verified. Preliminary verdicts and the final test case verdict can be seen in the log file.

